<commit_before>import org.jetbrains.dokka.gradle.DokkaMultiModuleTask

dependencies {
    implementation(kotlin("stdlib"))
}

val olderVersionsFolder = "olderVersions"

// The previously documentations should be generated with the versioning plugin
val generatePreviouslyDocTask by tasks.register<DokkaMultiModuleTask>("dokkaPreviouslyDocumentation") {
    dependencies {
        dokkaPlugin("org.jetbrains.dokka:all-modules-page-plugin:1.7.10")
        dokkaPlugin("org.jetbrains.dokka:versioning-plugin:1.7.10")
    }
    val configuredVersion = "0.9"
    outputDirectory.set(file(projectDir.toPath().resolve(olderVersionsFolder).resolve(configuredVersion)))
    pluginsMapConfiguration.set(mapOf("org.jetbrains.dokka.versioning.VersioningPlugin" to """{ "version": "$configuredVersion" }"""))
    addChildTasks(listOf(project("childProjectA"), project("childProjectB")), "dokkaHtmlPartial")
}

tasks.dokkaHtmlMultiModule {
    dependsOn(generatePreviouslyDocTask)
    val configuredVersion = "1.0"
    pluginsMapConfiguration.set(mapOf("org.jetbrains.dokka.versioning.VersioningPlugin" to """{ "version": "$configuredVersion", "olderVersionsDir": "$projectDir/$olderVersionsFolder" }"""))
}<commit_msg>Fix path in versioning example on Windows (#2632)<commit_after>import org.jetbrains.dokka.gradle.DokkaMultiModuleTask

dependencies {
    implementation(kotlin("stdlib"))
}

val olderVersionsFolder = "olderVersions"

// The previously documentations should be generated with the versioning plugin
val generatePreviouslyDocTask by tasks.register<DokkaMultiModuleTask>("dokkaPreviouslyDocumentation") {
    dependencies {
        dokkaPlugin("org.jetbrains.dokka:all-modules-page-plugin:1.7.10")
        dokkaPlugin("org.jetbrains.dokka:versioning-plugin:1.7.10")
    }
    val configuredVersion = "0.9"
    outputDirectory.set(file(projectDir.resolve(olderVersionsFolder).resolve(configuredVersion)))
    pluginsMapConfiguration.set(mapOf("org.jetbrains.dokka.versioning.VersioningPlugin" to """{ "version": "$configuredVersion" }"""))
    addChildTasks(listOf(project("childProjectA"), project("childProjectB")), "dokkaHtmlPartial")
}

tasks.dokkaHtmlMultiModule {
    dependsOn(generatePreviouslyDocTask)
    val configuredVersion = "1.0"
    pluginsMapConfiguration.set(mapOf("org.jetbrains.dokka.versioning.VersioningPlugin" to """{ "version": "$configuredVersion", "olderVersionsDir": "${projectDir.resolve(olderVersionsFolder).invariantSeparatorsPath}" }"""))
}
{'fraction_non_alphanumeric': 0.12869283690813435, 'fraction_numerical': 0.0113314447592068, 'mean_word_length': 6.62962962962963, 'pattern_counts': {'":': 6, '<': 5, '<?xml version=': 0, '>': 5, 'https://': 0, 'lorem ipsum': 0, 'www.': 0, 'xml': 0}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 0, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '6544808', 'n_tokens_mistral': 765, 'n_tokens_neox': 737, 'n_words': 99}